# All rights reserved.
#
# Redistribution and use in source and binary forms, with or without modification,
# are permitted provided that the following conditions are met:
#
# 1. Redistributions of source code must retain the above copyright notice, this
#    list of conditions and the following disclaimer.
#
# 2. Redistributions in binary form must reproduce the above copyright notice,
#    this list of conditions and the following disclaimer in the documentation
#    and/or other materials provided with the distribution.
#
# 3. Neither the name of the copyright holder nor the names of its contributors
#    may be used to endorse or promote products derived from this software without
#    specific prior written permission.
#
# THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" AND
# ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE IMPLIED
# WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE ARE
# DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE LIABLE FOR
# ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES
# (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES;
# LOSS OF USE, DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND ON
# ANY THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT
# (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF THIS
# SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.
#
#
#
# 2012-01-31, Lars Bilke
# - Enable Code Coverage
#
# 2013-09-17, Joakim Söderberg
# - Added support for Clang.
# - Some additional usage instructions.
#
# USAGE:

# 0. (Mac only) If you use Xcode 5.1 make sure to patch geninfo as described here:
#      http://stackoverflow.com/a/22404544/80480
#
# 1. Copy this file into your cmake modules path.
#
# 2. Add the following line to your CMakeLists.txt:
#      INCLUDE(CodeCoverage)
#
# 3. Set compiler flags to turn off optimization and enable coverage:
#    SET(CMAKE_CXX_FLAGS "-g -O0 -fprofile-arcs -ftest-coverage")
#	 SET(CMAKE_C_FLAGS "-g -O0 -fprofile-arcs -ftest-coverage")
#
# 3. Use the function SETUP_TARGET_FOR_COVERAGE to create a custom make target
#    which runs your test executable and produces a lcov code coverage report:
#    Example:
#	 SETUP_TARGET_FOR_COVERAGE(
#				my_coverage_target  # Name for custom target.
#				test_driver         # Name of the test driver executable that runs the tests.
#									# NOTE! This should always have a ZERO as exit code
#									# otherwise the coverage generation will not complete.
#				coverage            # Name of output directory.
#				)
#
# 4. Build a Debug build:
#	 cmake -DCMAKE_BUILD_TYPE=Debug ..
#	 make
#	 make my_coverage_target
#
#

# Check prereqs
FIND_PROGRAM( GCOV_PATH NAMES $ENV{GCOV} gcov )
FIND_PROGRAM( LCOV_PATH lcov )
FIND_PROGRAM( GENHTML_PATH genhtml )
FIND_PROGRAM( GCOVR_PATH NAMES $ENV{GCOVR} gcovr PATHS ${CMAKE_SOURCE_DIR}/tests)
FIND_PROGRAM( COVERALLS_PATH coveralls-lcov)

IF(NOT GCOV_PATH)
	MESSAGE(FATAL_ERROR "gcov not found! Aborting...")
ELSE()
  MESSAGE(STATUS "Found ${GCOV_PATH}")
ENDIF() # NOT GCOV_PATH

SET(ENABLE_COVERAGE_FOUND ON CACHE STRING "Code coverage support found.")

IF ( NOT (CMAKE_BUILD_TYPE STREQUAL "Debug" OR CMAKE_BUILD_TYPE STREQUAL "Coverage"))
	MESSAGE( WARNING "Code coverage results with an optimized (non-Debug) build may be misleading" )
ENDIF() # NOT CMAKE_BUILD_TYPE STREQUAL "Debug"

SET(CMAKE_CXX_FLAGS_COVERAGE
	"-O0 --coverage ${COVERAGE_DEBUG_FLAGS}"
	CACHE STRING "Flags used by the C++ compiler during coverage builds."
	FORCE )
SET(CMAKE_C_FLAGS_COVERAGE
	"-O0 --coverage ${COVERAGE_DEBUG_FLAGS}"
	CACHE STRING "Flags used by the C compiler during coverage builds."
	FORCE )
SET(CMAKE_EXE_LINKER_FLAGS_COVERAGE
	""
	CACHE STRING "Flags used for linking binaries during coverage builds."
	FORCE )
SET(CMAKE_SHARED_LINKER_FLAGS_COVERAGE
	""
	CACHE STRING "Flags used by the shared libraries linker during coverage builds."
	FORCE )
SET(COVERAGE_BINARY_DIR "${PROJECT_BINARY_DIR}/src")
SET(COVERAGE_BASE_DIR "${PROJECT_SOURCE_DIR}")
set(COVERAGE_INFO_FILE "${PROJECT_BINARY_DIR}/gen/coverage.info")
MARK_AS_ADVANCED(
	CMAKE_CXX_FLAGS_COVERAGE
	CMAKE_C_FLAGS_COVERAGE
	CMAKE_EXE_LINKER_FLAGS_COVERAGE
	CMAKE_SHARED_LINKER_FLAGS_COVERAGE 
	COVERAGE_SOURCE_DIR)

# Param TARGET     The name of new the custom make target
# Param OUTPUT     lcov output is generated as OUTPUT.info
#                       HTML report is generated in OUTPUT/index.html
# Optional fourth parameter is passed as arguments to _testrunner
#   Pass them in list form, e.g.: "-j;2" for -j 2
FUNCTION(SETUP_TARGET_FOR_COVERAGE)
	set(options "")
	set(singleValueArgs TARGET OUTPUT)
	cmake_parse_arguments(SETUP_TARGET_FOR_COVERAGE "${options}" "${singleValueArgs}" "${multiValueArgs}" ${ARGN})

	if (NOT SETUP_TARGET_FOR_COVERAGE_TARGET) 
		MESSAGE(FATAL_ERROR "No target specified to SETUP_TARGET_FOR_COVERAGE")
	endif()

	if (NOT SETUP_TARGET_FOR_COVERAGE_OUTPUT)
		MESSAGE(FATAL_ERROR "No output specified to SETUP_TARGET_FOR_COVERAGE")
	endif()

	IF(NOT LCOV_PATH)
		MESSAGE(FATAL_ERROR "lcov not found! Aborting...")
	ELSE()
		MESSAGE(STATUS "Found ${LCOV_PATH}")
	ENDIF() # NOT LCOV_PATH

	IF(NOT GENHTML_PATH)
		MESSAGE(FATAL_ERROR "genhtml not found! Aborting...")
	ELSE()
		MESSAGE(STATUS "Found ${GENHTML_PATH}")
	ENDIF() # NOT GENHTML_PATH

	if (NOT COVERALLS_PATH)
		MESSAGE(STATUS "Coveralls supported disabled")
	else()
		MESSAGE(STATUS "Coveralls support enabled")
	endif()

  ADD_CUSTOM_TARGET(${SETUP_TARGET_FOR_COVERAGE_TARGET}_clean
    # Cleanup lcov
    COMMENT "Cleaning coverage info"
    COMMAND ${LCOV_PATH} --gcov-tool ${GCOV_PATH} --directory ${COVERAGE_BINARY_DIR} --zerocounters
  )

  ADD_CUSTOM_TARGET(${SETUP_TARGET_FOR_COVERAGE_TARGET}_generate
  	COMMENT "Executing tests for coverage"
    COMMAND ${CMAKE_CTEST_COMMAND} --verbose ${ARGN}
  )

  ADD_CUSTOM_TARGET(${SETUP_TARGET_FOR_COVERAGE_TARGET}_generate_clean
	DEPENDS ${SETUP_TARGET_FOR_COVERAGE_TARGET}_clean
	COMMENT "Executing tests for coverage"
	COMMAND ${CMAKE_CTEST_COMMAND} --verbose ${ARGN}
  )

  ADD_CUSTOM_TARGET(${SETUP_TARGET_FOR_COVERAGE_TARGET}_info
	DEPENDS ${SETUP_TARGET_FOR_COVERAGE_TARGET}_generate_clean
	COMMENT "Gathering coverage info"
    # Capturing lcov counters and generating report
    COMMAND ${LCOV_PATH} --gcov-tool ${GCOV_PATH} -q --directory ${COVERAGE_BINARY_DIR} --base-directory ${COVERAGE_BASE_DIR} --no-external --capture --output-file ${COVERAGE_INFO_FILE}
  )

	if (COVERALLS_PATH)
		ADD_CUSTOM_TARGET(${SETUP_TARGET_FOR_COVERAGE_TARGET}_coveralls
			DEPENDS ${SETUP_TARGET_FOR_COVERAGE_TARGET}_info
			COMMAND ${COVERALLS_PATH} ${COVERAGE_INFO_FILE}
			WORKING_DIRECTORY ${COVERAGE_SOURCE_DIR}
			COMMENT "Processing code coverage counters and uploading report."
		)
	endif()

	# Setup target
	ADD_CUSTOM_TARGET(${SETUP_TARGET_FOR_COVERAGE_TARGET}

		DEPENDS ${SETUP_TARGET_FOR_COVERAGE_TARGET}_info

		COMMAND ${CMAKE_COMMAND} -E remove_directory ${SETUP_TARGET_FOR_COVERAGE_OUTPUT}
		COMMAND ${GENHTML_PATH} -o ${SETUP_TARGET_FOR_COVERAGE_OUTPUT} ${COVERAGE_INFO_FILE}
		COMMAND ${CMAKE_COMMAND} -E remove ${COVERAGE_INFO_FILE}

		WORKING_DIRECTORY ${COVERAGE_SOURCE_DIR}
		COMMENT "Processing code coverage counters and generating report."
		)

	# Show info where to find the report
	ADD_CUSTOM_COMMAND(TARGET ${SETUP_TARGET_FOR_COVERAGE_TARGET} POST_BUILD
		COMMAND ;
		COMMENT "Open ./${SETUP_TARGET_FOR_COVERAGE_OUTPUT}/index.html in your browser to view the coverage report."
		)

ENDFUNCTION() # SETUP_TARGET_FOR_COVERAGE
